
Abstract. Riveret et al. proposed a framework for probabilistic legal
reasoning. Their goal is to determine the chance of winning a court case,
given the chances of the judge accepting certain claims and legal rules.
We tackle the same problem by defining and implementing a new formal-
ism, called probabilistic argumentation logic (PAL). We implement PAL
in CHRiSM, and discuss how it can be seen as a probabilistic generaliza-
tion of Nute’s defeasible logic. Not only does this provide an automation
of the — only hand-performed — computations in Riveret et al, it also
provides a solution to one of their open problems: a method to determine
the initial probabilities from a given body of precedents.

1 Introduction

Riveret et al. [1] proposed a framework of probabilistic legal reasoning based on
the argumentation framework of Prakken et al. [2], which provides a dialectical
proof theory in the formal setting of Dung [3]. Their goal is to determine the
chance of winning a court case, given the probabilities of the judge accepting
certain claimed facts to be valid and legal rules to be applicable.

Roth et al. [4] tackle a similar problem, but with the focus on finding legal
strategies for the involved parties to maximize their chances of winning. They
propose a rather complex framework, based on a logic layer, an argument layer,
a dialectical layer, a procedural layer, and finally a probabilistic weighting.

To our knowledge, none of these approaches have been implemented so far.
Both papers only contain a hand-performed computation to illustrate their ap-
proach on an example. Such an implementation of an ‘argument assistance sys-
tem’ is left explicitly as future work in [1].

Another issue is how to determine or verify the probabilities, which are as-
sumed to be known in advance. Riveret et al. [1] suggest that maybe somehow a
statistical analysis of how the judge has decided in the past could be performed,
but they again leave this issue to future work.

In this paper we tackle the problem by defining a new formalism, called prob-
abilistic argumentation logic, which can be seen as a probabilistic generalization
of defeasible logic with explicit underlying assumptions.

We formalize our approach and implement it in CHRiSM [5]. CHRiSM is a
rule-based probabilistic logic programming language based on CHR [6] in the

host language PRISM [7]. For reasons of space we will omit an introduction to
these formalisms; we assume the reader is already familiar with them. We refer
the reader to [6, 8, 9] for CHR and to [5] for CHRiSM. In a companion paper
[10], an introduction to CHRiSM can be found, as well as a discussion of the
advantages and disadvantages of CHRiSM in this setting.

Our implementation provides an automation of the probability computations
that were hand-performed in [1]. The built-in learning algorithm of CHRiSM also
provides a solution to the open problem of determining the probabilities.

The remainder of this paper is organized as follows. In Section 2 we introduce
the approach of [1] and the running example in their paper. We show how this
example can be encoded in CHRiSM. Section 3 introduces our new formalism,
probabilistic argumentation logic. We show that it generalizes Nute’s defeasible
logic [11] and we give an implementation in CHRiSM. In Section 4 we briefly
discuss learning. Finally we conclude in Section 5.

2 The mad cow example and dialogue games

The running example in [1] is the following. John, the proponent, wants to sue
Henry, the opponent, claiming compensation for the damage that Henry’s cow
caused to him when he drove off the road to avoid the cow. John argues that an
animal’s owner has to pay damages caused by their animal, that Henry is the
owner of the cow, and that the accident was caused by the need to avoid the
cow (argument A). Henry can counterattack in various ways: he can claim that
the damage was due to John’s negligence (he did not pay sufficient attention to
crossing animals) – argument B – or that it was a case of force majeure: the cow
suddenly went crazy and crossed into the street – argument C. The last objection
could be replied to by using the debated rule that only exogenous events count
as force majeure, and the cow’s madness is endogenous (argument D).

[1] assumes an abstract argumentation framework [3], which consists of a set
of arguments and a binary “defeats” relation. They then define the notion of a
dialogue game which captures the rules of legal argumentation. A dialogue is a
sequence of abstract arguments, in which the proponent and the opponent alter-
nate, each argument defeats the previous argument, and the proponent cannot
repeat arguments and his arguments have to strictly defeat the previous argu-
ment of the opponent.

Moreover, each of the arguments has some given “construction chance”,
which is the probability that the judge will actually accept the argument. The
aim is to estimate the overall chance that the case is won.

2.1 Encoding of dialogue games in CHRiSM

In general, [1] assume an abstract argumentation framework [3], which consists of
a set of arguments and a binary “defeats” relation. Argument X strictly defeats
argument Y if X defeats Y and Y does not defeat X. For example, argument A
and B defeat one another, while argument D strictly defeats argument C.

Moreover, each of the arguments has some given “construction chance”,
which is the probability that the judge will actually accept the argument. The
aim is to estimate the overall chance that the case is won.

They consider dialogue games which capture the rules of legal argumentation.
A dialogue is a sequence of arguments, in which the proponent and the opponent
alternate, with the following conditions:

– The proponent cannot repeat arguments;
– Every opponent argument defeats the previous (proponent) argument;
– Every proponent argument strictly defeats the previous argument;
– If a player cannot make a move, he loses.

The first argument is the main claim of the proponent. If the proponent wins all
possible dialogues, he wins the case. Otherwise the opponent wins. Obviously,
the outcome depends on which arguments are actually accepted by the judge.

In the mad cow example there are two possible dialogues if the judge accepts
all arguments: [A,B] (in which the proponent loses) and [A,C,D] (in which the
proponent wins). So if the judge accepts all arguments, the proponent loses (since
he has no winning strategy in case the opponent uses argument B).

We can directly encode the rules of the dialogue game in CHRiSM, as follows.
We use a dummy constraint predicate begin/0 to initialize a dialogue:

begin <=> init_defeats, dialogue([]).

The auxiliary predicate init defeats/0 initializes the “defeats” relation:

init_defeats <=> defeats(argA,argB), defeats(argB,argA),

defeats(argA,argC), defeats(argC,argA),

defeats(argD,argC).

The “strictly defeats” relation is derived from the “defeats” relation:

defeats(A,B) ==> strictly_defeats(A,B).

defeats(B,A) \ strictly_defeats(A,B) <=> true.

The main constraint predicate is dialogue/1, which contains a (reversed)
list representing a (partial) dialogue. For example, a dialogue [A,C,D] would be
encoded as dialogue([p-argD,o-argC,p-argA]). Note the reversed order (for
easy access to the last element) and the “p-” and “o-” tags to denote the player.

First of all, we make sure that the construction chances are taken into ac-
count. If the construction chance of some argument is p, then we simply prune
away a partial dialogue that ends with that argument with probability 1−p. For
example, suppose the construction chance of argument A is 0.9, then we remove
a dialogue ending with A with probability 0.1. These are the values used in [1]:

0.1 ?? dialogue([_-argA|_]) <=> true.

0.6 ?? dialogue([_-argB|_]) <=> true.

0.9 ?? dialogue([_-argC|_]) <=> true.

0.85 ?? dialogue([_-argD|_]) <=> true.

Now we add a rule for the proponent to make his main claim, argument A:

dialogue([]) ==> dialogue([p-argA]).

The rest of the dialogue is constructed according to the above conditions:

dialogue([p-A|D]), defeats(B,A) ==> dialogue([o-B,p-A|D]).

dialogue([o-A|D]), strictly_defeats(B,A) ==>

\+ member(p-B,D) | dialogue([p-B,o-A|D]).

Now comes a tricky part: if a partial dialogue has been extended (but only
after it has been extended in all possible ways), we can discard it. If it could not
be extended, we have a final dialogue, which was won by the last player. One
way to implement this is as follows: (exploiting the refined operational semantics
and the passive pragma)

% dialogue was extended: remove prefix

dialogue([_|D])#passive \ dialogue(D) <=> true.

% dialogue was not extended: last player wins

dialogue([X-_|_]) ==> winner(X).

Finally, as soon as there is one dialogue which is won by the opponent, then
the proponent cannot be a winner.

winner(o) \ winner(p) <=> true.

In [1], several pages are used to calculate the probability that the proponent
wins the case. In CHRiSM, we can simply use the following query:

?- prob begin ===> winner(p).

Probability of begin ===> winner(p) is: 0.494100000000000

Where did the dialog go? In the encoding of the arguments in Section 2, we
abstract away the turn-based dialog. We impose no total order on the arguments,
nor strict alternation between the players; in fact, we do not even record which
party (proponent or opponent) makes which claim. The essential structure of
the reasoning is left intact though.

The rules were written as if both the proponent and the opponent make all
of their claims and arguments simultaneously at the start. In order to model
more accurately that some arguments will only be put forward as a “reaction”
to the acceptance of other statements, we can also write the rules in a different
way, for example, instead of always claiming that the cow was mad:

0.2 ?? begin ==> accept(e,[]).

the opponent Henry only makes that claim if the judge is tempted to make him
compensate John’s damages:

0.2 ?? accept(c,_) ==> accept(e,[]).

Similarly, instead of always insisting that cow madness is endogenous:

0.3 ?? begin ==> accept(f,[]).

the proponent John only makes that claim if the judge has actually accepted
that the cow was mad:

0.3 ?? accept(e,_) ==> accept(f,[]).

With respect to the outcome of the case (whether or not statement c gets ac-
cepted), there is no difference between both ways of encoding. The only difference
is that the “put everything on the table at once” approach can introduce irrele-
vant claims, while the “reactive” approach more closely models the dialog, since
claims are only triggered when they are needed.

2.2 CHRiSM encoding of the mad cow example

In the above, we described a CHRiSM encoding of dialogue games with abstract
arguments, in which the arguments are treated as black boxes that only interact
through the “defeats” relation. We now proceed with a finer level of granularity.

The arguments in [1] consist of premises and rules, which each have a prob-
ability of being accepted by the judge. For example, argument A consists of the
premises that Henry owns the cow (a), and that the accident was caused by the
need to avoid the cow (b), together with the rule “a ∧ b → c”, where c stands
for “Henry has to compensate damages”. If a will certainly be accepted, b is ac-
cepted with a probability of 0.9, and the rule “a ∧ b→ c” is certainly accepted,
then the overall construction chance of argument A is 1× 0.9× 1 = 0.9.

We will call both premises and conclusions “statements” and use ground
Prolog terms to denote them. The auxiliary predicate neg/2 simply negates a
literal in a way that avoids double negations.

We use a dummy predicate begin/0 (to be used as the initial goal). The main
constraint predicate is accept/2, which indicates that the judge conditionally
accepts some statement: accept(S,C) denotes that statement S is accepted if
all conditions C (a Prolog list of statements) hold. If C is the empty list, the
statement is unconditionally accepted; otherwise the acceptance of the statement
can still be retracted if one of the conditions turn out to be unacceptable. If a
statement is already accepted with conditions A, then it is redundant to also
accept it with stronger conditions B ⊇ A.

accept(X,A) \ accept(X,B) <=> subset(A,B) | true.

The above rule implies a set semantics for unconditionally accepted statements.
A condition is redundant if it is implied by the other conditions:

accept(A,B) \ accept(X,C) <=> select(A,C,D), subset(B,D) | accept(X,D).

If a statement Y was accepted with conditions B, but one of those conditions
is contradicted (“undercut”) by a statement X with weaker conditions A that
are implied by B, then the acceptance of Y has to be retracted — we use a
simpagation rule to remove the accept(Y,B) constraint:

accept(X,A) \ accept(Y,B) <=>

subset(A,B), neg(X,NX), member(NX,B) | true.

Finally, we allow no contradictions:

accept(X,A), accept(Y,B) ==> subset(A,B), neg(X,Y) | fail.

Now we encode the premises and the rules of the arguments:

% Argument A (rule r1, premises a and b):

% "If Henry is the owner of the cow (a) and the accident was caused by the

% need to avoid the cow (b), then Henry has to compensate damages (c)."

1.0 ?? accept(a,[]), accept(b,[]) ==> accept(c, [app(r1)]).

1.0 ?? begin ==> accept(a,[]).

0.9 ?? begin ==> accept(b,[]).

The rule being used is a defeasible rule, so its conclusion c will be accepted
with the condition that the rule is actually applicable (app(r1)). This condition
can be “undercut” by arguments B or C.

% Argument B (rule r2, premise d):

% "If John was negligent (d), then r1 is not applicable."

0.8 ?? accept(d,[]) ==> accept(not(app(r1)),[]).

0.5 ?? begin ==> accept(d,[]).

For example, the judge could accept a, b and d and both rules, to reach
the state “accept(c,[app(r1)]), accept(not(app(r1)),[])”. Now the un-
dercutting rule removes the conditional acceptance of c, because its condition
was contradicted.

% Argument C (rule r3, premise e):

% "If the cow was mad (e), it was a case of force majeure.

% so r1 is not applicable."

0.5 ?? accept(e,[]) ==> accept(not(app(r1)), [app(r3)]).

0.2 ?? begin ==> accept(e,[]).

Again, the above rule (r3) is defeasible, so its conclusion can only be accepted
with the condition app(r3), which can be undercut by the final argument:

% Argument D (rule r4, premise f):

% "If the cow’s madness is endogenous (f), then the ’force majeure’

% rule r3 is not applicable."

0.5 ?? accept(f,[]) ==> accept(not(app(r3)), []).

0.3 ?? begin ==> accept(f,[]).

The only thing now left to do, is to resolve the conditions by making assump-
tions. For example, one possible result of the query begin is the following:

accept(a, []), accept(b, []), accept(e, []),

accept(c, [app(r1)]), accept(not(app(r1)), [app(r3)])

In this case, both c and not(app(r1)) are conditionally accepted. Although
not(app(r1)) undercuts the conditions of c, the “undercut” rule is not appli-
cable (yet) because the condition [app(r3)] is not weaker than the condition
[app(r1)]. However, since there is no counter-evidence for app(r3), we can as-
sume this condition to hold, “promoting” not(app(r1)) to an unconditionally
accepted statement, which then causes the acceptance of c to be retracted.
To implement this idea, we add the following rules at the end of the program:

begin <=> assume.

assume, accept(X,C) ==> select(A,C,D), accept(A,[]) ; true.

assume <=> true.

The middle rule nondeterministically selects conditions and accepts them. This
can either lead to a contradiction (e.g. if there is counter-evidence), which causes
backtracking, or succeed.

This concludes our program. We can now compute the desired probability —
which took several pages of manual calculations in [1] — with a simple query:

?- prob begin ===> accept(c,[]).

Probability of begin ===> accept(c,[]) is: 0.494100000000000

3 Generalization and Formalization

In order to generalize the running example, we will propose a transformation
from an arbitrary probabilistic legal argumentation logic A — a notion that will
be introduced in this section — to a CHRiSM program PCHRiSM (A).

3.1 Probabilistic Argumentation Logic

We use lit(A) to denote the set of literals over a set of atomic formulas A, i.e.
lit(A) = A ∪ {¬a | a ∈ A}. We will sometimes denote conjunctions over literals
as sets since the order of the conjuncts is irrelevant.

Definition 1 (Probabilistic argumentation logic). A probabilistic argu-
mentation logic or PAL is a tuple (S,A,R, P ), where S is a set of statements
and A is a set of assumptions (with S ∩ A = ∅), R is a set of rules, and P is a
function assigning probabilities to each rule, P : R 7→ [0, 1]. The rules in R have
the following form:

s1 ∧ . . . ∧ sn ⇒ c1 ∧ . . . ∧ cm assuming a1 ∧ . . . ∧ ak

where the left hand side (the antecedent) is a conjunction of literals (si ∈ lit(S ∪
A)) which can be empty (n ≥ 0), the right hand side (the consequent) is a
non-empty (m ≥ 1) conjunction of literals (ci ∈ lit(S ∪ A), and the part after

“assuming” (the assumption) is a possibly empty (k ≥ 0) conjunction of assump-
tion literals (ai ∈ lit(A)). If the left hand side is empty, the rule is also called
a fact and the arrow can be omitted; if the part after “assuming” is empty, the
rule is called unconditional and the keyword “assuming” can be omitted.

To illustrate the definition, we now write out the running mad cow example
as a formal probabilistic argumentation logic

Amc := ({a, b, c, d, e, f}, {app(r1), app(r3)}, Rmc, Pmc)

where the rules Rmc = {r1, r2, r3, r4, sa, sb, sd, se, sf} are the following:

r1 := a ∧ b⇒ c assuming app(r1)

r2 := d⇒ ¬app(r1)
r3 := e⇒ ¬app(r1) assuming app(r3)

r4 := f ⇒ ¬app(r3)
∀x ∈ {a, b, d, e, f} : sx := x

and the probabilities Pmc are the following:

Pmc := {(r1, 1), (r2, 0.8), (r3, 0.5), (r4, 0.5), (sa, 1), (sb, 0.9), (sd, 0.5), (se, 0.2), (sf , 0.3)}

Now we define an interpretation of a PAL as a set of conditional statements.

Definition 2 (Conditional statement). Given a PAL A = (S,A,R, P ), a
conditional statement is a pair (s, C) with s ∈ lit(S ∪ A) and C ⊆ lit(A), such
that C is not self-contradictory, that is, there is no c ∈ C and ¬c ∈ C.

Definition 3 (Interpretation). Given a PAL A, an interpretation is a set I
of conditional statements of A such that if (s, C1) ∈ I and (¬s, C2) ∈ I, then
C1 6⊆ C2 and C2 6⊆ C1.

In other words, the conditional statements are not directly contradictory —
although both a statement and its negation can be conditionally accepted at the
same time, as long as the assumptions are different.

Definition 4 (Partial Ordering of Interpretations). We say an interpre-
tation I1 is smaller than an interpretation I2, denoted I1 ≤i I2, if for all con-
ditional statements (s, c1) ∈ I1, there is a corresponding conditional statement
(s, c2) ∈ I2 such that c1 ⊆ c2.

If I1 ≤i I2, then the set of statements in I1 is a subset of the statements in
I2, so I1 makes less claims than I2 (which is why we call it smaller), but the
claims in I1 are in a sense stronger since they have a weaker condition.

We now define the semantics of a PAL, somewhat inspired by the definitions
in [11], but extending them to take the explicit conditions into account, as well
as the rule selection (which will be needed to introduce the rule probabilities).

Definition 5 (Compliant Interpretation w.r.t. Rule Selection). Given a
PAL A = (S,A,R, P ) and a set of selected rules Rs ⊆ R, a compliant interpre-
tation I w.r.t. the selected rules Rs is a minimal (w.r.t. ≤i) interpretation that
satisfies the following additional criterion:

– if Rs contains a rule r = (Sr ⇒ Cr assuming Ar),
– and ∀s ∈ Sr : ∃c : (s, c) ∈ I (the antecedent is conditionally accepted),
– and ∀a ∈ Ar : ¬∃c : (¬a, c) ∈ I (the assumption is not questioned),
– then for every set {(s1, c1), . . . , (sn, cn)} ⊆ I such that Sr = s1 ∧ . . . ∧ sn, it

must hold that ∀x ∈ Cr : ∃y ⊆ Ar ∪ {c1, . . . , cn} : (x, y) ∈ I (the consequent
is conditionally accepted).

Definition 6 (Valid Interpretation w.r.t. Rule Selection). Given a PAL
A = (S,A,R, P ) and a set of selected rules Rs ⊆ R, a valid interpretation
I w.r.t. the selected rules Rs is a compliant interpretation without gratuitous
statements, that is, there is no subset K ⊆ I with K 6= ∅ such that:

– if Rs contains a rule r = (Sr ⇒ Cr assuming Ar),
– and ∀s ∈ Sr : ∃c : (s, c) ∈ I \K, and ∀a ∈ Ar : ¬∃c : (¬a, c) ∈ I,
– then for every set {(s1, c1), . . . , (sn, cn)} ⊆ I \K such that Sr = s1∧ . . .∧sn,

it must hold that ∀x ∈ Cr : ∀y ⊆ Ar ∪ {c1, . . . , cn} : (x, y) 6∈ K.

Returning to the mad cow example, consider the rule selection Rmc of all
rules. The following is the only valid interpretation w.r.t. Rmc:

{(a, ∅), (b, ∅), (d, ∅), (e, ∅), (f, ∅), (¬app(r1), ∅), (¬app(r3), ∅)}

This is the only valid interpretation w.r.t. R′ = {r1, r3, r4, sa, sb, se, sf}:

{(a, ∅), (b, ∅), (c, ∅), (e, ∅), (f, ∅), (¬app(r3), ∅)}

An interpretation like the above, but with (c, {app(r1)}) instead of (c, ∅) also
satisfies the criterion of Def. 5, but it is not compliant because it is not minimal
w.r.t. ≤i. The condition for the acceptance of the consequent of applicable rules
is allowed to be weaker than (i.e. a subset of) the union of all the conditions
arising from the antecedent and assumption. This relaxation serves two goals.
Firstly, it means that if a consequent can be derived in different ways such that it
would be accepted multiple times with varying conditions, it suffices to have only
the weakest conditions in the interpretation. Secondly, because the interpretation
has to be minimal w.r.t. ≤i, conditions will only be present in the interpretation
to avoid contradiction. The following example illustrates this point.

r1 := bird ⇒ flies assuming normal-bird

r2 := tux ⇒ bird ∧ tuxedo-plumage assuming eyes-OK

r3 := tuxedo-plumage⇒ penguin assuming feathers-make-bird

r4 := penguin⇒ ¬flies
r5 := tux

Consider the selection of all rules. The following interpretation is compliant:

{(tux, ∅), (¬eyes-OK, ∅)}

but it is not a valid interpretation since the statement ¬eyes-OK is trivially
gratuitous: there is not even a rule that could derive it.

The following interpretation satisfies the criterion of Def. 5:







(tux, ∅), (bird, {eyes-OK}), (flies, {eyes-OK, normal-bird}),
(tuxedo-plumage, {eyes-OK}), (penguin, {eyes-OK, feathers-make-bird}),

(¬flies, {eyes-OK, feathers-make-bird})







but it is not minimal w.r.t. ≤i; we can relax some conditions to get a minimal
interpretation, for example: (in this case there are three minimal interpretations)

{

(tux, ∅), (bird, ∅), (flies, {normal-bird}),
(tuxedo-plumage, ∅), (penguin, {eyes-OK}), (¬flies, {eyes-OK})

}

Note that some conditions have to be kept in order to avoid a direct con-
tradiction between flies and ¬flies. Also note that in the above program,
one could replace r4 with penguin⇒ ¬normal-bird,¬flies to get rid of these
conditions and have a unique minimal interpretation which contains (¬flies, ∅).

Definition 7 (Plausibility of a statement). Given a PAL A = (S,A,R, P ),
a statement literal s ∈ lit(S) is called plausible w.r.t. a rule selection Rs if all
valid interpretations w.r.t. Rs contain a conditional statement (s, c) for some c.

Definition 8 (Acceptability of a statement). Given a PAL A = (S,A,R, P ),
a statement literal s ∈ lit(S) is called acceptable w.r.t. a rule selection Rs if it
is plausible and there exists a valid interpretation w.r.t. Rs which contains the
conditional statement (s, ∅).

In the above example, tux, bird, and tuxedo-plumage are acceptable state-
ments, while flies, ¬flies, and penguin are plausible but not acceptable. All
other literals are not even plausible.

The probability prob(Rs) of a rule selection Rs ⊆ R is defined as follows:

prob(Rs) =

(

∏

r∈Rs

P (r)

)





∏

r∈R\Rs

1− P (r)





which ensures that
∑

Rs∈P(R) prob(Rs) = 1.

Definition 9 (Probability of a statement). Given a PAL A = (S,A,R, P ),
the probability of a statement s ∈ lit(S) is defined as the sum of the probabilities
of all rule selections Rs ∈ P(R) in which s is acceptable.

3.2 Transformation to CHRiSM

We now introduce a transformation from an arbitrary probabilistic legal ar-
gumentation logic A = (S,A,R, P ) to a CHRiSM program PCHRiSM (A). The
transformation is a generalization of the example discussed in Section 2.2. We
assume the list predicates member/2, subset/2, and append/2 (whose first ar-
gument is a list of lists) are already defined in the host language (Prolog).

The transformed program starts with the same three rules as in Section 2.2.
These rules insure that accept/2 encodes an interpretation, redundant condi-
tional statements are removed, as well as defeated statements. It ends with the
rules for the assume phase as in Section 2.2.

In between, there are two CHRiSM rules for each rule of A. Each rule r ∈ R:

s1 ∧ . . . ∧ sn ⇒ c1 ∧ . . . ∧ cm assuming a1 ∧ . . . ∧ ak

is transformed into one simple probabilistic CHRiSM rule:

P (r) ?? begin ==> selected(r).

and one non-probabilistic CHRiSM rule:

1 ?? selected(r), accept(s1,C1), . . ., accept(sn,Cn)

==> append([C1, . . ., Cn, [a1, . . ., ak] ], NC),

accept(c1,NC), . . ., accept(cm,NC).

where all literals of the form ¬x are encoded as not(x).

Correctness. It is relatively straightforward to see that when the assume phase
starts, accept/2 encodes an interpretation that satisfies the criterion of Def. 5
w.r.t. the rule selection encoded by selected/1. It also does not contain gra-
tuitous statements. However, the interpretation is not necessarily minimal. The
assume phase searches for minimal interpretations by nondeterministically re-
laxing the assumptions. Since the relaxed accept/2 constraints will cause the
transformed PAL rules to be revisited, the criterion of Def. 5 remains satisfied.

3.3 Relationship between PAL and Defeasible Logic

In defeasible logic [11], there are three kinds of rules: strict rules (denoted with
→), defeasible rules (denoted with ⇒), and undercutting defeaters (denoted
with ;). There is also a precedence relation ≺ over the non-strict rules, to
settle conflicting rules. Conflicts are defined using a set of conflict sets, which
should contain at least all sets of the form {φ,¬φ}.

Without loss of generality (cf. Theorem 6 of [12]) we can assume that the
precedence relation is empty, there are no undercutting defeaters, and the conflict
set are just the atomic formulas and their negations.

Given a closed defeasible theory D with a set of initial facts F and a set of
rules RS ∪ RD (where RS are the strict rules and RD are the defeasible rules),
we can translate it to a probabilistic argumentation logic PAL(D) = (S,A,R, P )
as follows:

– The set of statements S is the set of atoms appearing in F , RS and RD;
– The set of assumptions A consists of fresh symbols a1, . . . , a|RD|;
– The probability function P is the constant function 1;
– The rules R are constructed as follows:

• For every fact f ∈ F , we add a rule ⇒ f .
• For every strict rule (X → Y ) ∈ RS , we add a rule X ⇒ Y .
• For every defeasible rule (X ⇒ Y ) ∈ RD, we add a ruleX ⇒ Y assuming ai,
where i is the number corresponding to the defeasible rule.

• If the i-th and the j-th defeasible rule are conflicting rules, that is, if rule
i is of the form (X ⇒ ψ) ∈ RD and rule j is of the form (Y ⇒ ¬ψ) ∈ RD,
then we add the rules ψ ⇒ ¬aj and ¬ψ ⇒ ¬ai.

• If the i-th defeasible rule (X ⇒ ψ) ∈ RD conflicts with a strict rule
(Y → ψ̄) ∈ RS (where ψ̄ denotes the negation of the literal ψ), then we
add the rule ψ̄ ⇒ ¬ai

To illustrate this translation, let us look at a trivial example: the defeasible
rules ⇒ p and ⇒ ¬p. Since these rules contradict one another, neither p nor ¬p
can be inferred. The translated PAL rules are the following:

⇒ p assuming a1
⇒ ¬p assuming a2
p⇒ ¬a2
¬p⇒ ¬a1

This PAL has two valid interpretations (w.r.t. all rules): {(p, ∅), (¬a2)} and
{(¬p, ∅), (¬a1)}. As desired, neither p nor ¬p are acceptable.

Conjecture 1 (Weak Completeness). The above translation is complete in the
following sense: for all literals p in LitD, we have that if D |∼ p (p is defeasibly
derivable from D), then p is plausible in PAL(D) (w.r.t. the selection of all rules)
and if D ∼| p (p is defeasibly refutable in D), then p is not acceptable in PAL(D).

Conjecture 2 (Soundness). The above translation is sound in the following sense:
if the probability of p in PAL(D) is 1, then D |≈ p (p is defeasibly entailed by
D), and if the probability of p in PAL(D) is 0, then D |6≈ p (p is not defeasibly
entailed by D).

4 Learning

For now, we have assumed the probabilities to be known in advance. As men-
tioned in the conclusion of [1], an issue is: where do these numbers come from?
They suggest a statistical analysis of known precedents. The CHRiSM framework
gives us exactly the tools needed to do this.

Instead of using fixed probabilities, we can make some or all probabilities
learnable. We can then use a “training set” consisting of the outcomes of earlier
similar cases — we call these the observations — to find a maximum likelihood
probability distribution to fit the observations.

Table 1. Re-discovering the probabilities with CHRiSM’s learning algorithm.

Rule Orig. prob. Learned prob.

a 1 0.999999723
b 0.9 0.875267302
d 0.5 0.540243696
e 0.2 0.204545455
f 0.3 0.258644714

Rule Orig. prob. Learned prob.

r1 1 0.999975929
r2 0.8 0.723429684
r3 0.5 0.547720330
r4 0.5 0.503165649

Some of the observations may be full observations, meaning that we not
only know the final outcome, but also how exactly the reasoning went: what
statements were put forward, what statements were accepted or rejected. More
realistically, we only have partial observations: e.g. the final outcome is known,
but not the intermediate steps. In CHRiSM we can use both.

As a proof of concept, let us try to “rediscover” the original probabilities in
our running example. Assume for the sake of the example we have a database of
1100 prior rulings, but we do not have the time and resources to read through
all of them to find out exactly what the reasoning was. Say that we take 100
random samples and input them as full observations, for example

begin <==> accept(not app(r1),[]),accept(d,[]),accept(b,[]),accept(a,[]).

In this example observation, even though both a and b were accepted, c
was not accepted, either because rule r1 was not applied (remember, we do not
know that it should have probability 1), or because its assumption app(r1) was
refuted because d was accepted and rule r2 was applied.

The remaining 1000 cases are not studied in that much depth: all that was
checked is who won the case (i.e. was c accepted or not?) and whether or not
statement e (“the cow was mad”) was accepted. For the four possible combina-
tions, the following counts were recorded:

62 times begin ===> accept(c,[]), accept(e,[]).

432 times begin ===> accept(c,[]), ~accept(e,[]).

138 times begin ===> ~accept(c,[]), accept(e,[]).

368 times begin ===> ~accept(c,[]), ~accept(e,[]).

In full observations, the right hand side of the large double arrow has to be
exhaustive, so there is no need for explicit negation. In partial observations, the
rhs is not an exhaustive enumeration, so explicit negation (denoted by tilde) can
be useful, like in the above example.

The full observations above were obtained by simply taking 100 random
samples (i.e. running the query sample begin) on the original program with
explicit probabilities; the counts for the partial observations were obtained by
computing the probabilities for each case multiplying them by 1000.

Now that we have a training set, we can use the built-in learning algorithm to
find a probability distribution that fits the data. The resulting probabilities are
shown in Table 1; they approximate the original probabilities reasonably well.

5 Conclusion

We defined probabilistic argumentation logic (PAL) and showed how it can be
used for probabilistic legal reasoning in the style of [1]. We provided an imple-
mentation of PAL in CHRiSM through a straightforward encoding. Where pre-
vious work only showed by hand-performed examples how the problems could
be solved, we provide a formalism and its encoding in CHRiSM that implements
the problems and automates the generation of solutions. This automation goes
much further than the original problem statement. The resulting CHRiSM pro-
gram can be used to compute the probabilities of possible outcomes, to obtain
random samples, and to learn some or all underlying probabilities, solving an
open problem in [1].

It remains a topic for future work to relate PAL to other existing proposals
for defeasible reasoning, argumentation, and non-monotonic logic in general.
We have already indicated how Nute’s defeasible logic can be translated to a
fragment of PAL in a sound and weakly complete way, although we postpone
the proofs to future work.

9. Frühwirth, T., Raiser, F., eds.: Constraint Handling Rules: Compilation, Execu-
tion, and Analysis. BOD (March 2011)

